

12 Folder - Verstecken von Programmteilen

Fold

In der KR C1 ist es möglich einen KRL-Block für den Anwender auf der KUKA-Bedienoberfläche (KUKA-Bof) unsichtbar zu machen. Hierfür werden die betreffenden Vereinbarungen oder Anweisungen durch die Bezeichnungen ; **FOLD** und ; **ENDFOLD** eingeschlossen. Von der Sequenz

```
?
; FOLD RESET OUT

FOR I=1 TO 16
    SOUT[I]=FALSE
ENDFOR

; ENDFOLD
?
```

sind auf der Bedienoberfläche nur die Worte **RESET OUT** zu sehen.

Damit können Sie z.B. Deklarations- und Initialisierungsteil für den Anwender unsichtbar machen. Die Darstellung von Beispiel 11.1 auf der Bedienoberfläche ist in Abb. 59 zu sehen.



```
DEF FOLDS()

; FOLD DECLARATION; % weitere Informationen
;----- Deklarationsteil -----
EXT BAS (BAS_COMMAND :IN, REAL :IN )
DECL AXIS HOME
INT I
; ENDFOLD

; FOLD INITIALISATION
;----- Initialisierung -----
INTERRUPT DECL 3 WHEN $STOPMESS==TRUE DO IR_STOPM ( )
INTERRUPT ON 3
BAS (#INITMDV, 0 ) ; Initialisierung von Geschwindigkeiten,
; Beschleunigungen, $BASE, $TOOL, etc.

FOR I=1 TO 16
    SOUT[I]=FALSE
ENDFOR
HOME={AXIS: A1 0, A2 -90, A3 90, A4 0, A5 30, A6 0}
; ENDFOLD

;----- Hauptteil -----
PTP HOME ; SAK-Fahrt
LIN {X 540, Y 630, Z 1500, A 0, B 90, C 0}
PTP HOME

END
```



```

1
2  DECLARATION
3
4  INITIALISATION
5
6  ;----- Hauptteil -----
7  PTP  HOME ;SAK-Fahrt
8  LIN  {X 540,Y 630,Z 1500,A 0,B 90,C 0}
9  PTP  HOME
10
11  END
    
```

Abb. 59 Darstellung von Beispiel 11.1 auf der KUKA-Bof

Ein Fold kann auf Expertenebene im Menü "Bearbeiten" geöffnet oder geschlossen werden. Im geschlossenen Fold ist nur der Ausdruck nach dem Schlüsselwort **FOLD** sichtbar. Im geöffneten Fold sind dagegen alle Anweisungen und Vereinbarungen zu sehen.



FOLD ist lediglich ein Anweisung für den Editor. Der Compiler interpretiert die FOLD-Anweisungen aufgrund des vorangestellten Semikolons als normalen Kommentar.

Limited
Visibility

Ein weiteres Hilfsmittel um die Informationsmenge auf der Bedienoberfläche möglichst gering zu halten ist die Funktion "Limited Visibility" im Menü "Hilfe". Sie ist standardmäßig eingeschaltet und kann nur auf Expertenebene ausgeschaltet werden.

"Limited Visibility" unterdrückt z.B. die Zeile der Programmdefinition und alle Texte in einer **FOLD**-Zeile, welche nach den Zeichen ";" geschrieben werden. Diese Informationen werden aber zur Anzeige eines Inline-Formulars benötigt. Die Darstellung des obigen Programms auf der Bedienoberfläche mit ausgeschalteter "Limited Visibility"-Funktion ist in Abb. 60 zu sehen.

```

1  &ACCESS  RVO
2  DEF  FOLDER ( )
3
4  ;FOLD DECLARATION;% weitere Informationen
5
6  ;FOLD INITIALISATION
7
8  ;----- Hauptteil -----
9  PTP  HOME ;SAK-Fahrt
10  LIN  {X 540,Y 630,Z 1500,A 0,B 90,C 0}
11  PTP  HOME
12
13  END
    
```

Abb. 60 Darstellung von Beispiel 11.1 auf der KUKA-Bof mit ausgeschalteter "Limited Visibility"-Funktion



Erst wenn alle Folds geöffnet sind und "Limited Visibility" ausgeschaltet ist, sind dem Programmierer alle vorhandenen Programmzeilen verfügbar. Die Darstellung auf der Bedienoberfläche entspricht dann der Darstellung in einem normalen Texteditor.